NAG Fortran Library Routine Document

F08KCF (DGELSD)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F08KCF (DGELSD) computes the minimum-norm solution to a real linear least-squares problem

$$\min_x \|b - Ax\|_2.$$

2 Specification

```
      SUBROUTINE F08KCF (M, N, NRHS, A, LDA, B, LDB, S, RCOND, RANK, WORK,
     1                   LWORK, IWORK, INFO)
      INTEGER            M, N, NRHS, LDA, LDB, RANK, LWORK, IWORK(*), INFO
      double precision   A(LDA,*), B(LDB,*), S(*), RCOND, WORK(*)
```

The routine may be called by its LAPACK name ***dgelsd***.

3 Description

F08KCF (DGELSD) uses the singular value decomposition (SVD) of A. A is an m by n matrix which may be rank-deficient.

Several right-hand side vectors b and solution vectors x can be handled in a single call; they are stored as the columns of the m by r right-hand side matrix B and the n by r solution matrix X.

The problem is solved in three steps:

1. reduce the coefficient matrix A to bidiagonal form with Householder transformations, reducing the original problem into a 'bidiagonal least-squares problem' (BLS);
2. solve the BLS using a divide-and-conquer approach;
3. apply back all the Householder tranformations to solve the original least-squares problem.

The effective rank of A is determined by treating as zero those singular values which are less than RCOND times the largest singular value.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia URL: http://www.netlib.org/lapack/lug

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

1: M – INTEGER *Input*

On entry: m, the number of rows of A.

Constraint: $\mathrm{M} \geq 0$.

2: N – INTEGER *Input*

On entry: n, the number of columns of A.

Constraint: $\mathrm{N} \geq 0$.

3: NRHS – INTEGER *Input*

On entry: r, the number of right-hand sides, i.e., the number of columns of the matrices B and X.

Constraint: $\mathrm{NRHS} \geq 0$.

4: $\mathrm{A}(\mathrm{LDA},*)$ – ***double precision*** array *Input/Output*

Note: the second dimension of the array A must be at least $\max(1, \mathrm{N})$.

On entry: the m by n matrix A.

On exit: has been destroyed.

5: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F08KCF (DGELSD) is called.

Constraint: $\mathrm{LDA} \geq \max(1, \mathrm{M})$.

6: $\mathrm{B}(\mathrm{LDB},*)$ – ***double precision*** array *Input/Output*

Note: the second dimension of the array B must be at least $\max(1, \mathrm{NRHS})$.

On entry: the m by r right-hand side matrix B.

On exit: is overwritten by the n by r solution matrix X. If $m \geq n$ and $\mathrm{RANK} = n$, the residual sum-of-squares for the solution in the ith column is given by the sum of squares of elements $n+1, \ldots, m$ in that column.

7: LDB – INTEGER *Input*

On entry: the first dimension of the array B as declared in the (sub)program from which F08KCF (DGELSD) is called.

Constraint: $\mathrm{LDB} \geq \max(1, \max(\mathrm{M}, \mathrm{N}))$.

8: $\mathrm{S}(*)$ – ***double precision*** array *Output*

Note: the dimension of the array S must be at least $\max(1, \min(\mathrm{M}, \mathrm{N}))$.

On exit: the singular values of A in decreasing order.

9: RCOND – ***double precision*** *Input*

On entry: used to determine the effective rank of A. Singular values $\mathrm{S}(i) \leq \mathrm{RCOND} \times \mathrm{S}(1)$ are treated as zero.

If $\mathrm{RCOND} < 0$, ***machine precision*** is used instead.

10: RANK – INTEGER *Output*

On exit: the effective rank of A, i.e., the number of singular values which are greater than $\mathrm{RCOND} \times \mathrm{S}(1)$.

11: $\mathrm{WORK}(*)$ – ***double precision*** array *Workspace*

Note: the dimension of the array WORK must be at least $\max(1, \mathrm{LWORK})$.

On exit: if $\mathrm{INFO} = 0$, $\mathrm{WORK}(1)$ returns the optimal LWORK.

12: LWORK – INTEGER *Input*

On entry: the dimension of the array WORK as declared in the (sub)program from which F08KCF (DGELSD) is called.

The exact minimum amount of workspace needed depends on M, N and NRHS. As long as LWORK is at least

$$12 \times \mathrm{N} + 2 \times \mathrm{N} \times smlsiz + 8 \times \mathrm{N} \times nlvl + \mathrm{N} \times \mathrm{NRHS} + (smlsiz + 1)^2, \qquad \text{if } \mathrm{M} \geq \mathrm{N}$$

or

$$12 \times \mathrm{M} + 2 \times \mathrm{M} \times smlsiz + 8 \times \mathrm{M} \times nlvl + \mathrm{M} \times \mathrm{NRHS} + (smlsiz + 1)^2, \qquad \text{if } \mathrm{M} < \mathrm{N},$$

the code will execute correctly.

$smlsiz$ is equal to the maximum size of the subproblems at the bottom of the computation tree (usually about 25), and $nlvl = \max(0, \mathrm{int}(\log_2(\min(\mathrm{M},\mathrm{N})/(smlsiz + 1))) + 1)$.

For good performance, LWORK should generally be larger. Consider increasing LWORK by at least $nb \times \min(\mathrm{M},\mathrm{N})$, where nb is the optimal block size.

If LWORK $= -1$, a workspace query is assumed; the routine only calculates the optimal size of the array WORK and the minimum size of the array IWORK, and returns these values as the first entries of the WORK and IWORK arrays, and no error message related to LWORK is issued.

Constraint: LWORK must be at least 1.

13: IWORK($*$) – INTEGER array *Workspace*

Note: the dimension of the array IWORK must be at least $\max liwork$.

Constraint: $liwork \geq \max(1, 3 \times \min(\mathrm{M},\mathrm{N}) \times nlvl + 11 \times \min(\mathrm{M},\mathrm{N}))$, where $\min(\mathrm{M},\mathrm{N}) = \min(\mathrm{M},\mathrm{N})$.

On exit: if INFO $= 0$, IWORK(1) returns the minimum $liwork$.

14: INFO – INTEGER *Output*

On exit: INFO $= 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO $= -i$, the ith parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

The algorithm for computing the SVD failed to converge; if INFO $= i$, i off-diagonal elements of an intermediate bidiagonal form did not converge to zero.

7 Accuracy

See Section 4.5 of Anderson *et al.* (1999) for further details.

8 Further Comments

The complex analogue of this routine is F08KQF (ZGELSD).

9 Example

To solve the linear least-squares problem

$$\min_x \|b - Ax\|_2$$

for the solution, x, of minimum norm, where

$$A = \begin{pmatrix} -0.09 & -1.56 & -1.48 & -1.09 & 0.08 & -1.59 \\ 0.14 & 0.20 & -0.43 & 0.84 & 0.55 & -0.72 \\ -0.46 & 0.29 & 0.89 & 0.77 & -1.13 & 1.06 \\ 0.68 & 1.09 & -0.71 & 2.11 & 0.14 & 1.24 \\ 1.29 & 0.51 & -0.96 & -1.27 & 1.74 & 0.34 \end{pmatrix} \quad \text{and} \quad b = \begin{pmatrix} 7.4 \\ 4.3 \\ -8.1 \\ 1.8 \\ 8.7 \end{pmatrix}.$$

A tolerance of 0.01 is used to determine the effective rank of A.

Note that the block size (NB) of 64 assumed in this example is not realistic for such a small problem, but should be suitable for large problems.

9.1 Program Text

Note: the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     F08KCF Example Program Text
*     Mark 21 Release. NAG Copyright 2004.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          MMAX, NB, NLVL, NMAX
      PARAMETER        (MMAX=8,NB=64,NLVL=10,NMAX=16)
      INTEGER          LDA, LIWORK, LWORK
      PARAMETER        (LDA=MMAX,LIWORK=3*MMAX*NLVL+11*MMAX,
     +                 LWORK=NB*(2*MMAX+NMAX))
*     .. Local Scalars ..
      DOUBLE PRECISION RCOND
      INTEGER          I, INFO, J, LWRK, M, N, RANK
*     .. Local Arrays ..
      DOUBLE PRECISION A(LDA,NMAX), B(NMAX), S(MMAX), WORK(LWORK)
      INTEGER          IWORK(LIWORK)
*     .. External Subroutines ..
      EXTERNAL         DGELSD
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F08KCF Example Program Results'
      WRITE (NOUT,*)
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) M, N
      IF (M.LE.MMAX .AND. N.LE.NMAX .AND. M.LE.N) THEN
*
*        Read A and B from data file
*
         READ (NIN,*) ((A(I,J),J=1,N),I=1,M)
         READ (NIN,*) (B(I),I=1,M)
*
*        Choose RCOND to reflect the relative accuracy of the input
*        data
*
         RCOND = 0.01D0
*
*        Solve the least squares problem min( norm2(b - Ax) ) for the
*        x of minimum norm.
*
         CALL DGELSD(M,N,1,A,LDA,B,N,S,RCOND,RANK,WORK,LWORK,IWORK,
     +               INFO)
*
         IF (INFO.EQ.0) THEN
*
```

```
*           Print solution
*
            WRITE (NOUT,*) 'Least squares solution'
            WRITE (NOUT,99999) (B(I),I=1,N)
*
*           Print the effective rank of A
*
            WRITE (NOUT,*)
            WRITE (NOUT,*) 'Tolerance used to estimate the rank of A'
            WRITE (NOUT,99998) RCOND
            WRITE (NOUT,*) 'Estimated rank of A'
            WRITE (NOUT,99997) RANK
*
*           Print singular values of A
*
            WRITE (NOUT,*)
            WRITE (NOUT,*) 'Singular values of A'
            WRITE (NOUT,99999) (S(I),I=1,M)
         ELSE
            WRITE (NOUT,*) 'The SVD algorithm failed to converge'
         END IF
      ELSE
         WRITE (NOUT,*) 'MMAX and/or NMAX too small, and/or M.GT.N'
      END IF
      STOP
*
99999 FORMAT (1X,7F11.4)
99998 FORMAT (3X,1P,E11.2)
99997 FORMAT (1X,I6)
      END
```

9.2 Program Data

```
F08KCF Example Program Data

  5      6                                :Values of M and N

 -0.09  -1.56  -1.48  -1.09   0.08  -1.59
  0.14   0.20  -0.43   0.84   0.55  -0.72
 -0.46   0.29   0.89   0.77  -1.13   1.06
  0.68   1.09  -0.71   2.11   0.14   1.24
  1.29   0.51  -0.96  -1.27   1.74   0.34 :End of matrix A

  7.4
  4.3
 -8.1
  1.8
  8.7                                     :End of vector b
```

9.3 Program Results

```
 F08KCF Example Program Results

 Least squares solution
      1.5938    -0.1180    -3.1501     0.1554     2.5529    -1.6730

 Tolerance used to estimate the rank of A
      1.00E-02
 Estimated rank of A
      4

 Singular values of A
      3.9997     2.9962     2.0001     0.9988     0.0025
```